
Технология имитационного моделирования риска-анализа инвестиционного проекта методом Монте-Карло в среде EXCEL
Цель этого метода — множеством итераций получить набор значений результирующего показателя, для которых рассчитывается среднее, а также величина риска. 
Метод имитационного моделирования Монте-Карло создает дополнительную возможность при оценке риска за счет того, что делает возможным создание случайных сценариев (оптимистичного, пессимистичного, реалистичного). Применение анализа риска использует богатство информации (будь она в форме объективных данных или оценок экспертов) для количественного описания неопределённости, существующей в отношении основных переменных проекта и для обоснованных расчетов возможного воздействия неопределённости на эффективность инвестиционного проекта. Результат анализа риска выражается не каким-либо единственным значением финансового показателя, а в виде вероятностного распределения всех возможных значений этого показателя. Следовательно, потенциальный инвестор с помощью метода Монте-Карло будет обеспечен полным набором данных, характеризующих риск проекта. На этой основе он сможет принять взвешенное решение о предоставлении средств. Однако этот метод нельзя назвать универсальным, так как он предполагает точные входные данные, некореллируемость параметров и их нормальное распределение. 

Проведём имитационное моделирование анализа риска конкретного инвестиционного проекта.
Исходная информация: Фирма реализует инвестиционный проект по производству нового вида электрических розеток. В результате анализа технико-экономического обоснования проекта экспертами были выявлены три ключевых параметра проекта и определены возможные границы их изменений (табл. 1). Прочие параметры проекта считаются постоянными величинами (табл. 2).

Таблица 1. Ключевые параметры проекта

	Сценарий
	Показатели

	
	Пессимистичный
	Оптимистичный
	Реалистичный

	Объём выпуска, Q
	150
	300
	200

	Цена за штуку, P
	40
	55
	50

	Переменные затраты, V
	35
	25
	30


Таблица 2. Неизменяемые параметры проекта по производству электрических розеток

	Показатели
	Наиболее вероятное значение

	Постоянные затраты, F
	600

	Амортизация, A
	120

	Налог на прибыль, T
	60%

	Норма дисконта, r
	12%

	Срок проекта, n
	6

	Начальные инвестиции, IC
	2400


Первым этапом анализа, согласно сформулированному выше алгоритму, является определение зависимости результирующего показателя от исходных данных. При этом в качестве результирующего показателя обычно выступает один из критериев эффективности: чистая современная стоимость, NPV; индекс доходности, PI или внутренняя норма доходности, IRR. Выбираем первый показатель:

где n – срок проекта, r – норма дисконта,
[image: image2.wmf]t

NCF

 ─ величина чистого потока платежей в периоде t, IC – первоначальный объём инвестиций. 
Математическая модель рассматриваемого инвестиционного проекта будет выглядеть следующим образом

Тогда риск проекта рассчитывается как вероятность того, что чистая приведенная стоимость будет меньше нуля. 

Согласно исходным данным, значения нормы дисконта r и первоначального объёма инвестиций IC известны и считаются постоянными в течение срока реализации проекта. Ключевыми варьируемыми параметрами являются: переменные расходы V, объём выпуска Q и цена P. Диапазоны возможных изменений варьируемых показателей приведены в табл. 1. При этом исходим из предположения, что все ключевые переменные имеют равномерное распределение вероятностей.

Реализация третьего этапа может быть осуществлена только с применением ЭВМ, оснащенной специальными программными средствами. Проведение имитационных экспериментов в среде EXCEL можно осуществить двумя способами – с помощью встроенных функций и путем использования инструмента "Генератор случайных чисел" дополнения "Анализ данных". Рассмотрим оба способа и проведём их сравнительный анализ.
 1. Имитационное моделирование с применением встроенных функций  

Следует отметить, что применение встроенных функций целесообразно лишь в том случае, когда вероятности реализации всех значений случайной величины считаются одинаковыми. Тогда для имитации значений требуемой переменной можно воспользоваться математическими функциями СЛЧИС() или СЛУЧМЕЖДУ(). Функция СЛЧИС() возвращает равномерно распределенное случайное число E, большее, либо равное 0 и меньшее 1. Эта функция не имеет аргументов. Если установлен режим автоматических вычислений, принятый по умолчанию, то возвращаемый функцией результат будет изменяться всякий раз, когда происходит ввод или корректировка данных. В режиме ручных вычислений пересчет всей электронной таблицы (ЭТ) осуществляется только после нажатия клавиши [F9]. Настройка режима управления вычислениями производится установкой соответствующего флажка в подпункте "Вычисления" пункта "Параметры" темы "Сервис" главного меню.

Как следует из названия этой функции СЛУЧМЕЖДУ(нижн_граница; верхн_граница), она позволяет получить случайное число из заданного интервала. При этом тип возвращаемого числа зависит от типа заданных аргументов.

Итак, приступаем к разработке шаблона. С целью упрощения и повышения наглядности анализа выделим для его проведения в рабочей книге EXCEL два листа.

Первый лист – "Имитация", предназначен для построения генеральной совокупности (рис. 1).

Рис. 1. Лист “Имитация”

Определенные в данном листе формулы и собственные имена ячеек приведены в табл. 3.
Таблица 3. Формулы листа "Имитация".

	Ячейка
	Формула

	А10
	=СЛУЧМЕЖДУ($B$3;$C$3)

	B10
	=СЛУЧМЕЖДУ($B$4;$C$4)

	C10
	=СЛУЧМЕЖДУ($B$5;$C$5)

	D10
	=(B10*(C10-A10)-$F$3-$G$3)*(1-$H$3)+$G$3

	E10
	=ПС($F$5;$G$5;-D10)-$H$5


Формулы в ячейках А10:С10 генерируют значения для соответствующих переменных с учетом заданных в ячейках В3:С5 диапазонов их изменений. Формулы в ячейках D10:E10 вычисляют величину потока платежей и его чистую современную стоимость соответственно.


В ячейку В7 введем число 500, то есть требуемое число экспериментов. Формулы в ячейках A10:E10 растянем соответственно по столбцам до ячеек A500:E509. Результатом выполнения этих действий будет заполнение блока А10:Е509 случайными значениями 

ключевых переменных V, Q, P и результатами вычислений величин NCF (чистых поступлений от проекта) и NPV. Фрагмент результатов имитации, полученных нами, [image: image28.png]I3

приведен на рис. 2.

Рис. 2. Результаты имитации

Лист "Результаты анализа" (рис. 3) содержит функции, вычисляющие параметры распределения изменяемых (Q, V, P) и результатных (NCF, NPV) переменных и вероятности различных событий.  
[image: image29.png]MCXOAIHbIE yCoBMA JKCRepuMeHTa

2 Ay akcHay
3 |nepem.pacxons ] k3

4 |konmwectso 150 300

5 |uena il 55

6

7 |3kcnepumentos= 500 Homep crp.

[

nepementbie nocrynnenun

o ooy wonwecrso(@) | uewa (°) "I HCC(NPVE)
10 ] 185 0 720 560,217
11 il 292 51 25184 7954,17p.
12 31 242 a0 7032 491,14
13 % 238 a4 16456 395459
14 27 197 2 1014 176697,
15 Ell 263 53 25224 7148,33p.
16 p] 267 55 30% 10062, 23p
17 il 156 a 6432 244 46
16 K| 183 a7 6565 1122 88p,
1< "b W\ vwrawyan { Pesymerates aramvia [ farpamial £ fverpanwii [

FoToso

TocTonHHbIe 3aTpaThi—F AMOpTHIA

600
Cpok npoc
12%

Hopma uckonTa—r

<

UM





Рис.  3. Лист “Результаты анализа”

Определенные для данного листа формулы и собственные имена ячеек приведены в табл. 4. 
Таблица 4. Формулы листа "Результаты анализа"

	Ячейка
	Формула

	B8
	=СРЗНАЧ(Имитация!A10:A509)

	B9
	=СТАНДОТКЛОН.Г(Имитация!А10:А509)

	B10
	=B9/B8

	B11
	=МИН(Имитация!А10:А509)

	B12
	=МАКС(Имитация!А10:А509)

	С8
	=СРЗНАЧ(Имитация!B10:B509)

	C9
	=СТАНДОТКЛОН.Г(Имитация!B10:B509)

	C10
	=C9/C8

	C11
	=МИН(Имитация!B10:B509)

	C12
	=МАКС(Имитация!B10:B509)

	D8
	=СРЗНАЧ(Имитация!C10:C509)

	D9
	=СТАНДОТКЛОН.Г(Имитация!C10:C509)

	D10
	=D9/D8

	D11
	=МИН(Имитация!C10:C509)

	D12
	=МАКС(Имитация!C10:C509)

	E8
	=СРЗНАЧ(Имитация!D10:D509)

	E9
	=СТАНДОТКЛОН.Г(Имитация!D10:D509)

	E10
	=E9/E8

	E11
	=МИН(Имитация!D10:D509)

	E12
	=МАКС(Имитация!D10:D509)

	F8
	=СРЗНАЧ(Имитация!E10:E509)

	F9
	=СТАНДОТКЛОН.Г(Имитация!E10:E509)

	F10
	=F9/F8

	F11
	=МИН(Имитация!E10:E509)

	F12
	=МАКС(Имитация!E10:E509)

	           F13
	=СЧЁТЕСЛИ(Имитация!E10:Е509;"<0")

	           F14
	=СУММЕСЛИ(Имитация!E10:Е509;"<0")

	           F15
	=СУММЕСЛИ(Имитация!E10:Е509;">0")

	           E18
	=НОРМАЛИЗАЦИЯ(D18;$F$8;$F$9)

	           F18
	=НОРМСТРАСП(E18)


Рассмотрим значение функций, приведенных в таблице 4. Функции МИН() и МАКС() вычисляют минимальное и максимальное значение для массива данных из блока ячеек, указанного в качестве их аргумента.

Функция СЧЕТЕСЛИ() осуществляет подсчет количества ячеек в указанном блоке, значения которых удовлетворяют заданному условию. В данном случае, заданная в ячейке F13, эта функция осуществляет подсчет количества отрицательных значений NPV, содержащихся в блоке ячеек ЧСС.

Механизм действия функции СУММЕСЛИ() аналогичен функции СЧЕТЕСЛИ(). Отличие заключается лишь в том, что эта функция суммирует значения ячеек в указанном блоке, если они удовлетворяют заданному условию. В данном случае, заданные в ячейках F14.F15, функции осуществляет подсчет суммы отрицательных (ячейка F14) и положительных (ячейка F14) значений NPV, содержащихся в блоке ЧСС.

Две последние формулы (ячейки Е18 и F18) предназначены для проведения вероятностного анализа распределения NPV и требуют небольшого теоретического отступления. В рассматриваемом примере мы исходим из предположения о независимости и равномерном распределении ключевых переменных Q, V, P. Однако какое распределение при этом будет иметь результатная величина – показатель NPV, заранее определить нельзя. Одно из возможных решений этой проблемы – попытаться аппроксимировать неизвестное распределение каким-либо известным. При этом в качестве приближения удобнее всего использовать нормальное распределение. В прикладном анализе для целей аппроксимации широко применяется частный случай нормального распределения – стандартное нормальное распределение. Математическое ожидание стандартно распределенной случайной величины Е равно 0. Стандартное нормальное распределение характеризуется всего одним параметром – стандартным отклонением 

, равным 1.

Приведение случайной переменной E к стандартно распределенной величине Z осуществляется с помощью нормализации – вычитания средней и последующего деления на стандартное отклонение:

Как следует из предыдущей формулы, величина Z выражается в количестве стандартных отклонений. Для вычисления вероятностей по значению нормализованной величины Z используются специальные статистические таблицы. В EXCEL подобные вычисления осуществляются с помощью статистических функций НОРМАЛИЗАЦИЯ () и НОРМСТРАСП().
 Функция НОРМАЛИЗАЦИЯ(x;среднее;станд_откл) возвращает нормализованное значение Z величины x, на основании которого затем вычисляется искомая вероятность p. Функция требует задания трех аргументов: х (нормализуемое значение), среднее (математическое ожидание случайной величины Е), станд_откл (стандартное отклонение). Полученное значение Z является аргументом для следующей функции – НОРМСТРАСП(). Эта функция возвращает стандартное нормальное распределение, то есть вероятность того, что случайная нормализованная величина Е будет меньше или равна х. Она имеет всего один аргумент – Z, вычисляемый функцией НОРМАЛИЗАЦИЯ ().

Соответствующие проведенному эксперименту результаты анализа приведены на рис. 4.

Рис. 4. Результаты анализа

Рассмотрим результаты проведенных исследований. Коэффициент вариации NPV (=0,745865674, ячейка F10) меньше 1. Таким образом, риск данного проекта в целом ниже среднего риска инвестиционного портфеля фирмы. Результаты вероятностного анализа показывают, что вероятность получения отрицательной величины NPV составляет 0,090005042 (ячейка F18). Еще больший оптимизм внушают результаты анализа распределения чистых поступлений NCF от проекта. Величина стандартного отклонения здесь составляет всего 0,435027194 (ячейка E10) от среднего значения. Таким образом, с вероятностью более 0,9 можно утверждать, что поступления от проекта будут положительными величинами.

Сумма всех отрицательных значений NPV в полученной генеральной совокупности (ячейка F14) может быть интерпретирована как чистая стоимость неопределённости для инвестора в случае принятия проекта. Аналогично сумма всех положительных значений NPV (ячейка F15) может трактоваться как чистая стоимость неопределенности для инвестора в случае отклонения проекта. Несмотря на всю условность этих показателей, в целом они представляют собой индикаторы целесообразности проведения дальнейшего анализа. В данном случае они наглядно демонстрируют несоизмеримость суммы возможных убытков по отношению к общей сумме доходов (-24724,54 и 1673928,48 соответственно).

Подводя итоги можно отметить, что применение рассмотренной технологии проведения имитационных экспериментов в среде EXCEL – достаточно трудоемкий процесс, который, к тому же, ограничивается случаем равномерного распределения исследуемых переменных.

2. Имитационное моделирование с инструментом "Генератор случайных чисел"

Гораздо более удобным и эффективным способом решения таких задач в среде EXCEL является использование специального инструмента анализа – "Генератор случайных чисел".

Этот инструмент предназначен для автоматической генерации множества данных (генеральной совокупности) заданного объема, элементы которого характеризуются определенным распределением вероятностей. При этом могут быть использованы семь типов распределений: равномерное, нормальное, Бернулли, Пуассона, биномиальное, модельное и дискретное. Применение инструмента "Генератор случайных чисел", как и большинства используемых в этой работе функций, требует установки специального дополнения "Пакет анализа".

Для демонстрации техники применения этого инструмента изменим условия предыдущего примера, определив вероятности для каждого сценария развития событий следующим образом – см. табл. 5. Мы также будем исходить из предположения о нормальном распределении ключевых переменных. Количество имитаций оставим прежним – 500.

Таблица 5. Вероятностные сценарии реализации проекта

	Сценарий

показатели
	Пессимистичный

P=0,25
	Оптимистичный

P=0,25
	Реалистичный

P=0,5

	Объём выпуска, Q
	150
	300
	200

	Цена за штуку, P
	40
	55
	50

	Переменные затраты, V
	35
	25
	30


Приступим к формированию шаблона. Как и в предыдущем случае, выделим в рабочей книге два листа: "Имитация" и "Результаты анализа". Формирование шаблона целесообразно начать с листа "Результаты анализа" (рис. 5).

Рис. 5. Лист "Результаты анализа" (шаблон II)

Перейдем к следующему листу и присвоим ему имя – "Имитация". Приступаем к его формированию (рис.6).

Рис. 6. Лист "Имитация" (шаблон II)

Первая часть этого листа (блок ячеек А1:Е10) предназначена для ввода исходных данных и расчета необходимых параметров их распределений. Нормальное распределение случайной величины характеризуется двумя параметрами – математическим ожиданием (средним) и стандартным отклонением. Формулы расчета указанных параметров для ключевых переменных модели заданы в блоках ячеек В7:D7 и B8:D8 соответственно (см. табл. 6). Заполним ячейки соответственно формулам, указанным в таблице.

Таблица 6. Формулы листа "Имитация" (шаблон II)

	Ячейка
	Формула

	В7
	=СУММПРОИЗВ(B3:B5; E3:E5)

	В8
	{=КОРЕНЬ(СУММПРОИЗВ((B3:B5-B7)^2;E3:E5))}

	С7
	=СУММПРОИЗВ(C3:C5; E3:E5)

	С8
	{=КОРЕНЬ(СУММПРОИЗВ((C3:C5-C7)^2;E3:E5))}

	D7
	=СУММПРОИЗВ(D3:D5; E3:E5)

	D8
	{=КОРЕНЬ(СУММПРОИЗВ((D3:D5-D7)^2;E3:E5))}

	E10
	=B10+13 –1

	D13
	=(B13*(C13-A13)-'Результаты анализа'!$B$3-'Результаты анализа'!$B$4)*(1-'Результаты анализа'!$D$3)+'Результаты анализа'!$B$4

	E13
	=ПС('Результаты анализа'!$D$2;'Результаты анализа'!$D$4;-D13)-'Результаты анализа'!$B$2


Формула в ячейке Е10 по заданному числу имитаций (ячейка В10) вычисляет номер последней строки для блоков, в которых будут храниться сгенерированные значения ключевых переменных.


Ячейки D13.E13 содержат формулы для расчета величины потока платежей NCF и его чистой современной стоимости NPV.


Для проведения имитационного эксперимента, установим курсор в ячейку А13.

1. Выберем в главном меню "Данные" пункт "Анализ данных". Результатом выполнения этих действий будет появление диалогового окна "Анализ данных", содержащего список инструментов анализа.

2. Выберем из списка "Инструменты анализа" пункт "Генерация случайных чисел" и нажмем кнопку "ОК"

[image: image8.png]Avans ganne

Uncrpyments ananisa

[1eyX93KTOpHI AUCEPQHONHAI aHanWs Ges nosToperl +
Koppensuus.

Sxcnonenuaneoe cazksaHIE
ZeyesiGopossii F-recr ana gMcnepcii
anis Oypee

Tucrorpanmia

Gl i

Kosapuauus
Onucatensian craucrika !





3. На экране появится диалоговое окно "Генерация случайных чисел". Укажем в списке "Распределения" требуемый тип – "Нормальное". Заполним остальные поля изменившегося окна, согласно рис. 7, и нажмём кнопку "ОК". Результатом будет заполнение блока ячеек А13.А512 (переменные расходы) сгенерированными случайными значениями.

Рис. 7. Заполнение полей окна "Генерация случайных чисел"

Первым заполняемым аргументом диалогового окна "Генерация случайных чисел" является поле "Число переменных". Оно задает количество колонок электронной таблицы, в которых будут размещаться сгенерированные в соответствии с заданным законом распределения случайные величины. В нашей задаче оно должно содержать 1, так как ранее под значения переменной V (переменные расходы) в ЭТ была отведена одна колонка – колонка "А". В случае если указывается число большее 1, случайные величины будут размещены в соответствующем количестве соседних колонок, начиная с активной ячейки. Если это число не введено, то все колонки в выходном диапазоне будут заполнены.


Следующим обязательным аргументом для заполнения является содержимое поля "Число случайных чисел" (количество имитаций). Согласно условиям задачи, оно должно быть равно 500. При этом EXCEL автоматически подсчитывает необходимое количество ячеек для хранения генеральной совокупности.


Необходимый вид распределения задается путём соответствующего выбора из списка "Распределения". Как уже отмечалось ранее, могут быть получены семь наиболее распространенных в практическом анализе типов распределений, каждое из которых характеризуется собственными параметрами. Выбранный тип распределения определяет внешний вид диалогового окна. В рассматриваемом примере выбор типа распределения "Нормальное" повлёк за собой появление дополнительных аргументов – его параметров "Среднее" и "Стандартное отклонение", рассчитанных ранее для исследуемой переменной V в ячейках В7 и В8 листа "Имитация". 


Указание аргумента "Случайное рассеивание" позволяет при повторных запусках генератора получать те же значения случайных величин, что и при первом. Таким образом, одну и ту же генеральную совокупность случайных чисел можно получить несколько раз, что значительно повышает эффективность анализа. В случае если этот аргумент не задан (равен 0), при каждом последующем запуске генератора будет формироваться новая генеральная совокупность. В решаемой задаче этот аргумент задан равным единице, что позволит оперировать с одной и той же генеральной совокупностью и избежать постоянных перерасчетов ЭТ.


Последний аргумент диалогового окна "Генерация случайных чисел" – "Параметры вывода" определяет место расположения полученных результатов. Для проведения дальнейшего анализа необходимо, чтобы случайные величины размещались в специально отведенные для них блоки ячеек. В частности, для хранения 500 значений первой переменной ранее был отведен блок ячеек А13.А512. При увеличении либо уменьшении количества имитаций необходимо также переопределить и выходные блоки, предназначенные для хранения значений переменных.


Генерация значений остальных переменных Q и Р осуществляется аналогичным образом.


Для получения генеральной совокупности значений потока платежей и их чистой современной стоимости необходимо растянуть ячейки D13:E13 до D512:E512 соответственно.  

Рассмотрим результаты решения, рис. 8.

Рис. 8. Результаты имитационного эксперимента (шаблон II)

Заполним лист «Результаты анализа» (табл. 7) формулами. Ячейки B8:F15 заполняются так же, как и в предыдущем примере, а остальные заполним соответственно таблице.

Таблица 7. Формулы листа "Результаты анализа" (шаблон II)

	Ячейка
	Формула

	B17
	=НОРМРАСП(0;B8;B9;1)

	B18
	=НОРМРАСП(B11;B8;B9;1)

	B19
	=НОРМРАСП(B12;B8;B9;1)-НОРМРАСП(B8+B9;B8;B9;1)

	B20
	=НОРМРАСП(B8;B8;B9;1)-НОРМРАСП(B8-B9;B8;B9;1)

	C17
	=НОРМРАСП(0;C8;C9;1)

	C18
	=НОРМРАСП(C11;C8;C9;1)

	C19
	=НОРМРАСП(C12;C8;C9;1)-НОРМРАСП(C8+C9;C8;C9;1)

	C20
	=НОРМРАСП(C8;C8;C9;1)-НОРМРАСП(C8-C9;C8;C9;1)

	D17
	=НОРМРАСП(0;D8;D9;1)

	D18

	=НОРМРАСП(D11;D8;D9;1)

	D19
	=НОРМРАСП(D12;D8;D9;1)-НОРМРАСП(D8+D9;D8;D9;1)

	D20
	=НОРМРАСП(D8;D8;D9;1)-НОРМРАСП(D8-D9;D8;D9;1)

	E17
	=НОРМРАСП(0;E8;E9;1)

	E18
	=НОРМРАСП(E11;E8;E9;1)

	E19
	=НОРМРАСП(E12;E8;E9;1)-НОРМРАСП(E8+E9;E8;E9;1)

	E20
	=НОРМРАСП(E8;E8;E9;1)-НОРМРАСП(E8-E9;E8;E9;1)

	F17
	=НОРМРАСП(0;F8;F9;1)

	F18
	=НОРМРАСП(F11;F8;F9;1)

	F19
	=НОРМРАСП(F12;F8;F9;1)-НОРМРАСП(F8+F9;F8;F9;1)

	F20
	=НОРМРАСП(F8;F8;F9;1)-НОРМРАСП(F8-F9;F8;F9;1)


Рассмотрим результаты решения, рис.9.

Рис. 9. Результаты анализа (шаблон II)

Результаты проведенного имитационного эксперимента отличаются от предыдущих. Величина ожидаемой NPV равна 3429,280307 (ячейка F8) при стандартном отклонении 2868,035963 (ячейка F9). Коэффициент вариации (=0,836337571, ячейка F10) несколько выше, но меньше 1. Таким образом, риск данного проекта в целом ниже среднего риска инвестиционного портфеля фирмы. Результаты вероятностного анализа показывают, что вероятность получения отрицательной величины NPV составляет 0,115908887 (ячейка F17). Общее число отрицательных значений NPV в выборке составляет 53 (ячейка F10) из 500. Таким образом, с вероятностью 0,894 можно утверждать, что чистая современная стоимость проекта будет больше 0. При этом вероятность того, что величина NPV окажется больше чем М(NPV)+

, равна 0,158655149 (ячейка F19). Вероятность попадания значения NPV в интервал [М(NPV)

-

; М(NPV)] равна 0,341344746 (ячейка F20).


В анализе стохастических процессов большое значение имеют статистические взаимосвязи между случайными величинами. В качестве количественных характеристик подобных взаимосвязей в статистике используют два показателя: ковариацию и корреляцию.


Ковариация выражает степень статистической зависимости между двумя множествами данных и определяется из следующего соотношения:

где X, Y – множества значений случайных величин размерности m; а M(·) – математическое ожидание соответствующей случайной величины.


Положительная ковариация наблюдается в том случае, когда большим значениям случайной величины Х соответствуют большие значения случайной величины Y, то есть, когда между ними существует тесная прямая взаимосвязь. Соответственно отрицательная ковариация будет иметь место при соответствии малым значениям случайной величины Х больших значений случайной величины Y. При слабо выраженной зависимости значение показателя ковариации близко к нулю.


Ковариация зависит от единиц измерения исследуемых величин, что ограничивает ее применение на практике. Более удобным для использования в анализе является производный от нее показатель – коэффициент корреляции R, вычисляемый по формуле:

где знаменатель представляет собою произведение среднеквадратичных отклонений рассматриваемых случайных величин.


Коэффициент корреляции обладает теми же свойствами, что и ковариация, однако является безразмерной величиной и принимает значения от -1 (характеризует линейную обратную взаимосвязь) до +1 (характеризует линейную прямую взаимосвязь). Для независимых случайных величин значение коэффициента корреляции близко к 0.


Определим степень тесноты взаимосвязей между переменными V, Q, P, NCF и NPV. При этом в качестве меры будем использовать показатель корреляции R.

1. Выберем в главном меню тему "Данные" пункт "Анализ данных". Результатом выполнения этих действий будет появление диалогового окна "Анализ данных", содержащего список инструментов анализа. Выберем из списка "Инструменты анализа" пункт "Корреляция" и нажмем кнопку "ОК". Результатом будет появление окна диалога инструмента "Корреляция" (рис. 10).

2. Заполним поля диалогового окна, как показано на рис. 10, и нажмём кнопку "ОК".

Рис. 10. Заполнение окна диалога инструмента "Корреляция"

[image: image17.png]0582295884 0636307

Al B I ¢ [ b | E | )
1 nepem.pacxodul _Komuwecmeo __ugwa __nocmynnewus | HCC
2 [nepem.pac 1
3 |konuuects -0 034553344 1
4 |uewa 0019190079 -0,033989767 1
5 |noctynnen  -0,437591615 0582295884 0636307
B |4CC -0,437591615
7
8|




Рис. 11. Результаты корреляционного анализа

Результаты корреляционного анализа представлены в ЭТ в виде квадратной матрицы, заполненной только наполовину, поскольку значение коэффициента корреляции между двумя случайными величинами не зависит от порядка их обработки. Нетрудно заметить, что эта матрица симметрична относительно главной диагонали, элементы которой равны 1, так как каждая переменная коррелирует сама с собой.

Как следует из результатов корреляционного анализа, выдвинутая в процессе решения предыдущего примера гипотеза о независимости распределений ключевых переменных V, Q, P в целом подтвердилась. Значения коэффициентов корреляции между переменными расходами V, количеством Q и ценой Р (ячейки В3.В4, С4) достаточно близки к 0. В свою очередь, величина показателя NPV напрямую зависит от величины потока платежей (R = 1). Кроме того, существует корреляционная зависимость средней степени между Q и NPV (R≈0,58), P и NPV (R≈0,64). Как и следовало ожидать, между величинами V и NPV существует умеренная обратная корреляционная зависимость (R≈–0,44). Полезность проведения последующего статистического анализа результатов имитационного эксперимента заключается также в том, что во многих случаях он позволяет выявить некорректности в исходных данных, либо даже ошибки в постановке задачи. В частности, в рассматриваемой задаче отсутствие взаимосвязи между переменными затратами V и объемами выпуска продукта Q требует дополнительных объяснений, так как с увеличением последнего, величина V также должна расти. Таким образом, установленный диапазон изменений переменных затрат V нуждается в дополнительной проверке и, возможно, корректировке.

При проведении имитационного эксперимента и последующего вероятностного анализа полученных результатов мы исходили из предположения о нормальном распределении исходных и выходных показателей. Вместе с тем, справедливость сделанных допущений, по крайней мере, для выходного показателя NPV, нуждается в проверке.

В простейшем случае для этих целей можно использовать такие характеристики распределения, как асимметричность (коэффициент асимметрии или скоса) и эксцесс. Для нормального распределения эти характеристики должны быть равны 0. На практике близкими к нулевым значениям можно пренебречь. Для вычисления коэффициента асимметрии и эксцесса в EXCEL реализованы специальные статистические функции – СКОС и ЭКСЦЕСС. 

Но для более полного анализа полученных данных воспользуемся инструментом анализа данных «Описательная статистика».

Чем больше характеристик распределения случайной величины известно, тем точнее можно судить об описываемых ею процессах. Инструмент "Описательная статистика" автоматически вычисляет наиболее широко используемые в практическом анализе характеристики распределений. При этом значения могут быть определены сразу для нескольких исследуемых переменных. Определим параметры описательной статистики для переменных V, Q, P, NCF, NPV. Для этого выполним следующие шаги.

1. Выберем в главном меню тему "Сервис" пункт "Анализ данных". Результатом выполнения этих действий будет появление диалогового окна "Анализ данных", содержащего список инструментов анализа. Выберем из списка "Инструменты анализа" пункт "Описательная статистика".

2. Заполним поля диалогового окна, как показано на рис. 11 и нажмём кнопку "ОК"

Рис. 11. Заполнение полей диалогового окна "Описательная статистика"

Результатом выполнения указанных действий будет формирование отдельного листа, содержащего вычисленные характеристики описательной статистики для исследуемых переменных, рис.12.

Рис. 12. Описательная статистика для исследуемых переменных

Многие из приведенных в данной ЭТ характеристик нам уже хорошо знакомы, а их значения уже определены с помощью соответствующих функций на листе "Результаты анализа".


Вторая строка ЭТ содержит значения стандартных ошибок e для средних величин распределений. Другими словами, среднее или ожидаемое значение случайной величины М(Е) определено с погрешностью ± e. 


Как и математическое ожидание, медиана
 является одной из характеристик центра распределения случайной величины. В симметричных распределениях значение медианы должно быть равным или достаточно близким к математическому ожиданию. Как следует из полученных результатов, данное условие соблюдается для исходных переменных V, Q, P (значения медиан лежат в диапазоне М(Е) ± e, то есть   практически совпадают со средними). Однако для результатных переменных NCF, NPV значения медиан лежат ниже средних, что наводит на мысль о правосторонней асимметричности их распределений.


Мода – наиболее вероятное значение случайной величины (наиболее часто встречающееся значение в интервале данных). Для симметричных распределений мода равна математическому ожиданию. Иногда мода может отсутствовать. Если EXCEL вернул сообщение об ошибке, то вычисление моды не представляется возможным.


Эксцесс характеризует остроконечность (положительное значение) или пологость (отрицательное значение) распределения по сравнению с нормальной кривой. Теоретически эксцесс нормального распределения должен быть равен 0. Однако на практике для генеральных совокупностей больших объёмов его малыми значениями можно пренебречь. В рассматриваемой задаче примерно одинаковый положительный эксцесс наблюдается у всех переменных Q, NCF, NPV, V, P. Таким образом, графики этих распределений будут остроконечнее, по сравнению с нормальной кривой. 


Асимметричность (коэффициент асимметрии или скоса) характеризует смещение распределения относительно математического ожидания. При положительном значении коэффициента распределение скошено вправо, то есть его более длинная часть лежит правее центра (математического ожидания) и обратно. Для нормального распределения коэффициент асимметрии равен 0. На практике, его малыми значениями можно пренебречь. В частности, асимметрию распределений переменных V, Q, P в данном случае можно считать несущественной, чего нельзя, однако, сказать о распределении величины NPV.

Осуществим оценку значимости коэффициента асимметрии для распределения NPV. Наиболее простым способом получения такой оценки является определение стандартной (средней квадратической) ошибки асимметрии, рассчитываемой по формуле:

где n – число значений случайной величины (в данном случае – 500), s – коэффициент асимметрии.
Если отношение коэффициента асимметрии s к величине ошибки

), то асимметрия считается несущественной, а её наличие объясняется воздействием случайных факторов. В противном случае асимметрия статистически значима, и факт ее наличия требует дополнительной интерпретации. Осуществим оценку значимости коэффициента асимметрии для рассматриваемой ситуации. Введем в ячейку ЭТ формулу: 

= 0,61 / КОРЕНЬ(6*499 / 501*503) (Результат: 5,6).

Так как 

, то асимметрию следует считать существенной. Таким образом, первоначальное предположение о правосторонней скошенности распределения NPV подтвердилось. 

Для рассматриваемого примера наличие правосторонней асимметрии может считаться положительным моментом, так как это означает, что большая часть распределения лежит выше математического ожидания, то есть большие значения NPV являются более вероятными.

Оставшиеся показатели описательной статистики (рис. 44) представляют меньший интерес. Величина "Интервал" определяется как разность между максимальным и минимальным значением случайной величины (численного ряда). Параметры "Счёт" и "Сумма" представляют собой число значений в заданном интервале и их сумму соответственно.

Последняя характеристика "Уровень надёжности" показывает величину доверительного интервала для математического ожидания согласно заданному уровню надежности или доверия. По умолчанию уровень надежности принят равным 95%.

Для рассматриваемого примера это означает, что с вероятностью 0,95 величина математического ожидания NPV попадет в интервал 3429,28 ± 252,3. Следует отметить, что чем выше принятый уровень надёжности, тем больше будет величина доверительного интервала для среднего.

3. Сравнительный анализ результатов имитационных экспериментов

Теперь сравним результаты проведенных исследований. Коэффициент вариации NPV в первом случае приближённо равен 0,75 (рис.35, ячейка F10), 0,75<1. Это свидетельствует о том, что риск данного проекта в целом ниже среднего риска инвестиционного портфеля фирмы. Во втором случае указанный коэффициент немного выше (≈0,84) (рис.40, ячейка F10), но, все же, меньше 1, значит, тоже меньше среднего риска фирмы. Результаты вероятностного анализа в первом случае показывают, что вероятность получения отрицательной величины NPV составляет 0,09, во втором же случае эта величина несколько больше ─ 0,116. На практике столь небольшой разницей пренебрегать нельзя, ведь разница даже в 2,5%, может повлечь огромные финансовые потери для фирмы-инвестора. Таким образом, при решении рассматриваемой задачи целесообразнее воспользоваться вторым способом расчёта риска. 

Обоснованный выбор того или иного способа оценки степени риска в инвестиционных проектах должен базироваться на анализе исходных данных и учитывать   сильные и слабые стороны этих способов.

Положительные аспекты способа встроенных функций Excel:

· относительная простота вычислений даже для неопытного пользователя;

· наглядность результатов; 

· отсутствие необходимости знаний о вероятностных значениях распределения генерируемых параметров.

Отрицательные моменты способа встроенных функций Excel:

· продолжительное время работы;

· возможность значительных погрешностей (из-за использования для генерации величин только равномерного распределения).

Положительные аспекты способа «Генератор случайных чисел»: 

· высокая точность результатов (из-за возможности использования одного из семи основных распределений случайных величин).


Подводя итоги, можно сказать, что способ встроенных функций Excel целесообразно использовать только в том случае, когда необходимо лишь примерное значение оценки риска, и, когда амплитуда колебания переменных параметров проекта стремится к минимуму. В остальных же случаях целесообразнее использовать способ «Генератор случайных чисел».

4. Задание для выполнения. Информация о проекте

По состоянию на текущий момент основным видом деятельности предприятия-инициатора является оказание дополнительных услуг сотовой связи – предоставление информационного контента потребителям по их запросу. 

На основе анализа рынка мобильного контента разработчиками проекта определены пессимистический, наиболее вероятный и оптимистический варианты значений переменных, в соответствии с которыми устанавливаются диапазоны изменения риск-переменных используемых в работе моделей (табл. 1). Неизменяемые параметры проекта представлены в таблице 2.

Таблица 1

Таблица 2. Неизменяемые параметры проекта

	Показатели
	Наиболее вероятное значение

	Постоянные затраты в квартал, Ft
	2500000

	Амортизация в квартал, At
	700000

	Налог на прибыль, T
	20%

	Норма дисконта, r
	12%

	Инфляция, i 
	11%

	Срок проекта, n
	5

	Начальные инвестиции, IC
	12000000


Анализ инвестиционного проекта сроком реализации 5 лет провести для двух различных структур финансирования: стопроцентного собственного финансирования; финансирования с привлечением заемных средств.

Финансирование проекта за счет собственных средств 
Основные взаимосвязи в модели определяются следующим образом. Выручка от продаж определяется объемом оказанных услуг и ценой одной услуги. Издержки зависят от объема оказанных услуг, переменных издержек на оказание одной услуги и постоянных затрат. Переменные издержки на оказание одной услуги определяются затратами на рекламу для повышения спроса на услуги и затратами на оплату труда в расчете на одну оказанную услугу. Квартальный платеж зависит от выручки от оказанных услуг, издержек производства, амортизации, ставки налога на прибыль организаций, а также инвестиционных затрат. 

Учитывая изложенное, квартальный платеж может быть вычислен следующим образом:

CFt = (Qt ·pt – Qt · (mt + wt) – Ft – At) · (1 – T) + At – It,


 (2)

где CFt –платеж в t-й квартал; Qt – объем оказанных услуг в натуральном выражении в t-й квартал; pt – цена одной услуги в t-й квартал; mt – издержки на рекламу в расчете на одну оказанную услугу в t-й квартал; wt – затраты на оплату труда в расчете на одну оказанную услугу в t-й квартал; Ft – постоянные издержки в t-й квартал; At –амортизационные отчисления в t-й квартал; T – ставка налога на прибыль; It – инвестиционные затраты в t-й квартал. 

Для анализа и оценки рисков (возможных убытков) проекта будем использовать моделирование показателя чистого приведенного дохода инвестиционного проекта. 

Математическая модель рассматриваемого инвестиционного проекта будет выглядеть следующим образом:

(3)

где к переменным величинам модели отнесем Qt – объем продаж в натуральном выражении в t-й квартал; рt – цена за единицу продукции в t-й квартал; mt – переменные материальные затраты на единицу продукции в t-й квартал; 

Постоянными величинами (параметрами модели) будем считать I0 – начальные инвестиционные затраты; j – номинальную ставку дисконтирования; wt – затраты на оплату труда на одну оказанную услугу в t-й квартал; Ft – постоянные издержки в t-й квартал; At – амортизационные отчисления в t-й квартал; T – ставку налога на прибыль. 

При этом номинальная ставка дисконтирования может быть найдена по формуле 

1 + j = (1 + r) · (1 + i), 







(4)

где j – номинальная ставка дисконтирования; r – обычная ставка дисконтирования; i – темп инфляции. 

На основе модели по равномерному закону сгенерировать 500 значений для каждой риск-переменной и получить соответственно 500 значений результирующего показателя (NPV). Оценить качество модели.

Финансирование проекта с привлечением заемных средств 
Для этого в модели необходимо учесть условия кредитования: соотношение собственных и заемных средств 30:70; сумма основного долга 32 000 000 руб.; процентная ставка – 15 % годовых; погашение суммы основного долга равными долями в течение 8 последних кварталов реализации инвестиционного проекта. 

При этом изменение основных взаимосвязей в модели коснется только квартального платежа, который, помимо всего прочего, будет зависеть от процентов по кредиту, суммы полученных заемных средств и погашения суммы основного долга. 
Учитывая изложенное, квартальный платеж можно вычислить таким образом: 

CFt = (Qt ·pt – Qt · (mt + wt) – Ft – At - Dt) · (1 – T) + At – It +Zt - Lt, 

(5) 

где Dt – сумма выплачиваемых процентов по кредиту в t-й квартал; Zt – полученные заемные средства в t-й квартал; Lt – погашение суммы основного долга в t-й квартал.

Математическая модель для осуществления имитационного моделирования будет выглядеть следующим образом:

[image: image26.png]- CF,_ _; _

7§(1+j/4)' &
:f(Q.'PﬁQ.'(m,+W,)*1".*A.*D,)'(1*T)+4+Z.*L,71
=1 A+j/9)" °







(6)

Для данной схемы финансирования процесс генерации и последующего анализа остается прежним.

Провести сравнительный анализ результатов при 100%-ом финансировании за счет собственных средств и финансировании проекта с привлечением заемных средств.
� Медиана – это значение случайной величины, которое делит площадь, ограниченную кривой распределения, пополам (т.е. середина численного ряда или интервала).

